content: filter(xmasIsSunday, enumFromTo(2008, 2121))
  
  
  -- xmasIsSunday :: Int -> Bool
  on xmasIsSunday(y)
      tell (current date)
          set {its year, its month, its day, its time} to {y, 12, 25, 0}
          return its weekday is Sunday
      end tell
  end xmasIsSunday
  
  
  -- GENERIC FUNCTIONS ----------------------------------------------------------
  
  -- enumFromTo :: Int -> Int -> [Int]
  on enumFromTo(m, n)
      if m > n then
          set d to -1
      else
          set d to 1
      end if
      set lst to {}
      repeat with i from m to n by d
          set end of lst to i
      end repeat
      return lst
  end enumFromTo
  
  -- filter :: (a -> Bool) -> [a] -> [a]
  on filter(f, xs)
      tell mReturn(f)
          set lst to {}
          set lng to length of xs
          repeat with i from 1 to lng
              set v to item i of xs
              if |λ|(v, i, xs) then set end of lst to v
          end repeat
          return lst
      end tell
  end filter
  
  -- Lift 2nd class handler function into 1st class script wrapper
  -- mReturn :: Handler -> Script
  on mReturn(f)
      if class of f is script then
          f
      else
          script
              property |λ| : f
          end script
      end if
  end mReturn
  